// Appendix3.2
log using Appendix3.2.log, replace
use "C:\Users\sbstjp\OneDrive - Cardiff University\anes_timeseries_2024_stata_20250219.dta" // ANES 2024 Time Series Study // Preliminary Release: Pre-Election Data February 19, 2025 version

// Create social justice scale
* Rename variables 
rename V241290x Edi 
rename V241372x Tgbathroom 
rename V241375x Tgsport 
rename V241412x Appprotestgaza 

* Delete missing values
foreach var in Edi Tgbathroom Tgsport Appprotestgaza {
    replace `var' = . if `var' < 0
    tabulate `var', missing
}

*Reverse coding so social justice values are high
foreach var in Appprotestgaza Tgbathroom Edi {
    qui sum `var'
    local max_value = r(max)
    gen r`var' = `max_value' + 1 - `var'
}

*Standardize items in the scale from 1-2 - this avoids 0, for reasons outlined in next step
foreach var in rEdi rTgbathroom Tgsport rAppprotestgaza {
    summarize `var'
    gen s`var' = 1 + (`var' - r(min)) / (r(max) - r(min))
}

*At this point, the scale has a Cronbach's alpha of 0.76. 

* Replace missing values with 0 for the specified variables - this is necessary as Stata doesn't add up missing values and means a 0-1 standardization scale isn't feasible as missing values would overlap with the scale
foreach var in srEdi srTgbathroom sTgsport srAppprotestgaza {
    replace `var' = 0 if missing(`var')
}

* Initialize the total score and the count of non-zero responses
gen total_scoreSJV = 0
gen count_nonzeroSJV = 0

* Add each variable to the total scale score and count it if non-zero
foreach var in srEdi srTgbathroom sTgsport srAppprotestgaza  {
    replace total_scoreSJV = total_scoreSJV + `var'
    replace count_nonzeroSJV = count_nonzeroSJV + (`var' != 0)
}

* Calculate the average score, avoiding division by zero
gen socJusValues = .
replace socJusValues = total_scoreSJV / count_nonzeroSJV if count_nonzeroSJV > 0

// Liberalism scale
* Rename
rename V241269x bordersecurity
rename V241302 abortion
rename V241272x crime
rename V241308x deathpenalty
rename V241395x wall
rename V241389x birthrightcit
rename V241381x gayadoption
rename V241258 environment

*Drop missing values
foreach var in gayadoption bordersecurity crime deathpenalty wall birthrightcit {
    replace `var' = . if inlist(`var', -1, -2)
}

replace abortion = . if inlist(abortion, -9, -8, -1, 5)   
replace environment = . if inlist(environment, -9, -8, -1, 99)  

*Reverse code so liberal values are high
foreach var in gayadoption environment {
    qui sum `var'
    local max_value = r(max)
    gen r`var' = `max_value' + 1 - `var'
}
 
*Standardize items in the scale from 1-2 - this avoids 0, for reasons outlined in next step
foreach var in bordersecurity abortion crime deathpenalty wall birthrightcit rgayadoption renvironment {
    summarize `var'
    gen s`var' = 1 + (`var' - r(min)) / (r(max) - r(min))
}

* Replace missing values with 0 for the specified variables - this is necessary as Stata doesn't add up missing values and means a 0-1 standardization scale isn't feasible as missing values would overlap with the scale
foreach var in sbordersecurity sabortion scrime sdeathpenalty swall sbirthrightcit srgayadoption srenvironment {
    replace `var' = 0 if missing(`var')
}

* Initialize the total score and the count of non-zero responses
gen total_scoreAL = 0
gen count_nonzeroAL = 0

* Add each variable to the total scale score and count it if non-zero
foreach var in sbordersecurity sabortion scrime sdeathpenalty swall sbirthrightcit srgayadoption srenvironment {
    replace total_scoreAL = total_scoreAL + `var'
    replace count_nonzeroAL = count_nonzeroAL + (`var' != 0)
}

* Calculate the average score, avoiding division by zero
gen libValues = .
replace libValues = total_scoreAL / count_nonzeroAL if count_nonzeroAL > 0

// Life satisfaction
*Rename and delete missing values
rename V241621 lifesat 
replace lifesat = . if inlist(lifesat, -9, -8, -5, -1)

*now reverse, so happiness is coded high
egen maxval = max(lifesat)
gen reversed_lifesat = maxval + 1 - lifesat
drop maxval

*Standardize lifesat from 1-2, so it's like other dependent variables in the book
foreach var in reversed_lifesat {
    summarize `var'
    gen s`var' = 1 + (`var' - r(min)) / (r(max) - r(min))
}

rename sreversed_lifesat LifeSat

// Demographics
*Drop missing values and rename
replace V241177=. if V241177==99
replace V241177=. if V241177<0
rename V241177 libconsp

rename V241566x income 
replace income=. if income<0

replace V241491=. if V241491<0
rename V241491 unemployment

rename V241458x age 
replace age=. if age<0
egen age_centered = mean(age) 
replace age_centered = age - age_centered
gen age_centered_squared = age_centered^2

*Create dummies
gen FemaleGender=.
replace FemaleGender=1 if V241551==1
replace FemaleGender=2 if V241551==2

egen maxval = max(unemployment)
gen Unemployed = maxval + 1 - unemployment
drop maxval 

gen ReligiousAtt=. 
replace ReligiousAtt=0 if V241441==-1
replace ReligiousAtt=1 if inlist(V241441, 1, 2)

gen Graduate=.
replace Graduate=0 if inrange(V241465x, 1, 3)
replace Graduate=1 if inlist(V241465x, 4, 5)

gen Married=.
replace Married=1 if V241461x==1
replace Married=0 if inrange(V241461x, 2, 5)

// Standardize variables 
egen Age = std(age_centered_squared) 
egen Income = std(income)
egen LibValues = std(libValues)
egen SocJusValues = std(socJusValues)

// Regressions

regress LifeSat SocJusValues [pweight=V240105a], robust 
eststo
regress LifeSat SocJusValues Age FemaleGender Graduate Income Married ReligiousAtt Unemployed [pweight=V240105a], robust 
eststo
regress LifeSat LibValues [pweight=V240105a], robust 
eststo
regress LifeSat LibValues Age FemaleGender Graduate Income Married ReligiousAtt Unemployed [pweight=V240105a], robust 
eststo
regress LifeSat SocJusValues LibValues Age FemaleGender Graduate Income Married ReligiousAtt Unemployed [pweight=V240105a], robust 
eststo
esttab
 
log close

eststo clear

esttab using "Appendix3.2.rtf", ///
    b(3) se(3) ///
    star(* 0.05 ** 0.01 *** 0.001) ///
    coeflabels(_cons "Constant") ///
    nonumbers replace

// Alternative specification

ologit LifeSat SocJusValues Age FemaleGender Graduate Income Married ReligiousAtt Unemployed [pweight=V240105a], robust 


